

La función Lógica SI. Ejercicios Resueltos

Ec. HL. Mata¹

Es una de las funciones más potentes de Excel, la cual trabaja bajo el entorno: **Verdadero** y **Falso**. Decide el contenido de una celda dependiendo de que la prueba lógica sea cierta o falsa

Sintaxis:

=SI(Prueba_lógica;Valor_si_verdadero;Valor_si_falso)

Argumentos de la función:



La prueba lógica es cualquier valor o expresión que se pueda evaluar como **Verdadero** o como **Falso**.

Valor_si_verdadero es el valor que se devolverá si prueba lógica es **Verdadero**

Valor_si_falso es el valor que se devolverá si prueba lógica es **Falsa**

Escriba los argumentos secuencialmente, entre paréntesis, separándolos con puntos y comas y sin espacios en blanco.

En su forma más simple la función =**SI()** prueba el valor de una celda y ejecuta una acción. Si la prueba es positiva escribe (**Verdadero**); si es negativa escribe **Falso**.

Reglas para el uso de las funciones de Excel:

1. Si la función va al comienzo de una fórmula debe empezar por el signo =.
2. Los argumentos o valores de entrada van siempre entre paréntesis.
3. No dejes espacios antes o después de cada paréntesis.
4. Los argumentos pueden ser valores constantes (número o texto), fórmulas o funciones.
5. Coloque los argumentos literales (valores que vayan a ser mostrados como texto) entre comillas. No utilice comillas en los números
6. Los argumentos deben separarse por un punto y coma (;)

¹ Profesor Titular de la Facultad de Ciencias Económicas y Sociales (FACES) de la Universidad de los Andes (ULA). No hay ninguna pretensión de originalidad en estas notas. Las mismas existen por todas partes. Mi mayor contribución, si acaso alguna, consistió en ubicarlas, sistematizarlas, adaptarlas y publicarlas para beneficio de los Estudiantes de la Facultad de Ciencias Económicas y Sociales de la Universidad de los Andes.

Ejemplo A:

En este Ejemplo la función SI evaluará cada una de las celdas del rango A2:A9 para conocer si su contenido es Igual 0 (prueba lógica). Si se cumple dicha prueba, la función SI escribirá la palabra **Verdadero**, en el rango B2:B9 donde corresponda; de lo contrario, escribirá la palabra **Falso**

Prueba_lógica	A2=0
Valor_si_verdadero	Verdadero
Valor_si_falso	Falso

1. Entren a MS Excel. Hagan clic en la celda A1 para activarla y transcriban los siguientes datos, tal y como se indican en las columnas A y B, respectivamente:

	A	B
1	Valores	Acciones
2	12	
3	0	
4	13	
5	14	
6	0	
7	0	
8	0	
9	18	

2. Hagan clic en la celda B2 y ejecuten las siguientes instrucciones:

- Hagan clic en la herramienta $f(x)$. También pueden hacer clic en el menú **Insertar** y seleccionar el comando **Función**
- En el cuadro resultante, hagan clic en el panel de la izquierda de nombre **Categoría de la función y** seleccione **Lógicas**
- En el panel de la derecha, **Nombre de la función**, seleccionen la función **SI**
- Hagan clic en el botón de comando **Aceptar**

Cuadros de texto en el interior de la ventana de la función

SI

Prueba_lógica	<input type="text" value="A2=0"/>	=	valor_lógico
Valor_si_verdadero	<input type="text" value="Verdadero"/>	=	cualquiera
Valor_si_falso	<input type="text" value="Falso"/>	=	cualquiera

3. Hagan clic en la flecha de color rojo situada en el extremo derecho del cuadro de texto **Prueba_lógica**, para ocultarlo. Seleccionen la celda A2.
4. Hagan clic en la flecha roja del cuadro de diálogo **Prueba_lógica**, el cual se encuentra por encima de las etiquetas. Cuando regresen al cuadro coloquen el cursor a la derecha de la celda **A2** y escriban la prueba para la evaluación, es decir: escriba **=0** y opriman la tecla **Tab**. (la prueba dirá: **Si A2 es igual a cero**)
5. Hagan clic en el interior del cuadro de diálogo de nombre **Valor_si_verdadero** y escriban el argumento **Verdadero** y opriman la tecla **Tab**

6. Hagan clic en el interior del cuadro de diálogo de nombre **Valor_si_falso** y escriban el argumento **Falso**
7. Hagan clic en el botón de comando **Aceptar**
8. Coloquen el cursor nuevamente sobre la celda en la cual se colocó la acción, es decir en a celda **B2**
9. Pongan el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula almacenada en la celda **B2** en el resto del rango **B3:B9**

Análisis:

- Excel revisó la celda A2 para ver si su contenido era igual o diferente de cero (esta es la **Prueba lógica**). Puesto que el valor almacenado en la celda A2 era diferente de cero, la función **SI** ignoró la acción **Verdadero**, y mostró en su lugar la acción **Falso** (la acción correcta porque el valor almacenado en la celda A2 es diferente de cero).
- Excel revisó la celda A3 para ver si su contenido era igual o diferente de cero (la **Prueba lógica**). Puesto que el valor almacenado en la celda A3 era igual a cero, la función **SI** mostró la acción **Verdadero**, e ignoró la acción **Falso**.

Simulación:

1. Hagan clic sobre la celda A3 para activarla y escriban sobre ella el valor 10. Noten que la entrada en la celda A3 cambia instantáneamente de Verdadero a Falso, puesto que el contenido de dicha celda cambió de cero a 10.
2. Hagan clic nuevamente sobre la celda A3 y escriban ahora el valor cero. Observen que la función devuelve el resultado Verdadero

La función anterior también se puede escribir de la siguiente manera:

=SI(A2=0;"Verdadero";"Falso")

Guardar el libro

1. Hagan clic en el menú **Archivo** y seleccionen el comando **Guardar** o **Guardar Como**²
2. Despliegan la lista desplegable **Guardar en** y seleccionen **Disco de 3½ en A:**
3. En la lista desplegable **Nombre de Archivo**, escriban un nombre para guardar el Libro, escriban por ejemplo: la letra inicial de su nombre, la letra inicial de su apellido, la letra inicial del programa con el cual están trabajando y comiencen a numerar consecutivamente los archivos hechos con dicho programa, a partir de 01
4. Hagan clic en el botón de comando **Aceptar**

² Cuando se guarda un archivo por primera vez es indiferente utilizar el comando Guardar o Guardar Como

Ejemplo B:

Calculen las comisiones a pagar a 9 empleados de una determinada empresa en un todo de acuerdo con el siguiente plan: Cancele una comisión del 10 % a cada vendedor que venda Bs. 50000 o más durante el mes; de lo contrario, la comisión a pagar será solo del 2%.

<i>Prueba_lógica</i>	Ventas mensuales iguales o mayores a 50000
<i>Valor_si_verdadero</i>	Comisión de 10% para cada vendedor que venda 50000 ó mas
<i>Valor_si_falso</i>	Comisión del 2% para cada vendedor que venda 50000 ó menos

1. Activen la **Hoja2**. Hagan clic en la celda A1 para activarla y transcriban los siguientes datos, tal y como se indica en las columnas A y B, respectivamente:

	A	B
1	Ventas	Comisiones
2	75000	
3	45000	
4	48000	
5	55000	
6	100000	
7	35800	
8	120000	
9	80000	
10	49999	

2. Hagan clic en la celda B2 y escriban la siguiente función SI:

=SI (A2>=50000;A2*0,10;A2*0,02)

Argumentos de la función:

<i>Prueba_lógica:</i>	A2>=50000;
<i>Valor_si_verdadero:</i>	A2*0,10;
<i>Valor_si_falso:</i>	A2*0,02

3. Lleven el cursor nuevamente sobre la celda **B2**
4. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **B3:B10**. Actualicen el Libro.

	A	B
1	Ventas	Comisiones
2	75000	7500
3	45000	900
4	48000	960
5	55000	5500
6	100000	10000
7	35800	716
8	120000	12000
9	80000	8000
10	49999	999,98

Ejemplo C:

El siguiente ejercicio calcula un descuento del 15% (0,15) para aquellos clientes que paguen de **Contado**.

Prueba_lógica Pago de **Contado**
Valor_si_verdadero **Descuento del 15%** para quien pague al contado
Valor_si_falso Los que paguen a **Crédito no obtendrán descuento:0**

1. Activen la **Hoja3** y transcriban los siguientes datos, tal y como se indica en las columnas A, C y D, respectivamente:

	A	B	C	D
1			Subtotal	75000
2			Descuento	
3				
4	Forma de pago:			
5				
6				
7				
8				

2. Hagan clic en la celda A5 y escriban la etiqueta **Contado**
3. Hagan clic en la celda D2 y escriban manualmente la siguiente función SI:

=SI(A5="Contado";D1*0,15;0)

Argumentos de la función:

Prueba_lógica: **A5="Contado"**
Valor_si_verdadero: **D1*0,15;**
Valor_si_falso: **0**

5. Opriman la tecla **Enter** y actualicen el libro

La fórmula mirará si en la celda A5 (celda con contorno) existe la palabra **Contado**. En tal caso, ejecutará la fórmula (15 % de descuento de 75000), en caso contrario, colocará un cero en la celda D2, es decir, resultado **Falso**

	A	B	C	D
1			Subtotal	75000
2			Descuento	11250
3				
4	Forma de pago:			
5	Contado			
6				
7				
8				

Ejemplo D:

En el siguiente ejercicio la función SI evaluará cada una de las celdas del rango A2:A8 para ver si se cumple la siguiente condición: Si el valor de una determinada celda es Igual o inferior a cierto número, 16, por ejemplo, la función colocará en la celda la **Sumatoria** del Rango analizado; de lo contrario calculará y colocará el promedio del rango

Prueba_lógica A2<=16
Valor_si_verdadero Sumatoria del rango seleccionado, es decir, A2:A8
Valor_si_falso Promedio del rango seleccionado , es decir, A2:A8

1. Inserten una nueva Hoja de Excel, la cuarta, y transcriban los siguientes datos, tal y como se indica en la columnas A y B, respectivamente:

	A	B
1	Valores	Acciones
2	10	
3	12	
4	14	
5	16	
6	18	
7	20	
8	22	

4. Hagan clic en la celda B2 para activarla y escriban manualmente la siguiente función SI:

=SI(A2<=16;SUMA(\$A\$2:\$A\$8);PROMEDIO(\$A\$2:\$A\$8))

Argumentos de la función:

Prueba_lógica: A2<=16
Valor_si_verdadero: SUMA(\$A\$2:\$A\$8)
Valor_si_falso: PROMEDIO(\$A\$2:\$A\$8))

6. Opriman **Enter**. Copien la fórmula en el resto del rango y actualicen el libro

La fórmula mirará si en la celda A2 está almacenado el valor 16. En ese caso, ejecutará la fórmula (SUMA(\$A\$2:\$A\$8), de lo contrario, calculará y colocará el PROMEDIO(\$A\$2:\$A\$8)

	A	B
1	Valores	Acciones
2	10	112
3	12	112
4	14	112
5	16	112
6	18	16
7	20	16
8	22	16

Ejemplo E:

En el siguiente ejercicio la función SI inspeccionará el contenido de la celda A2. Si el valor encontrado es un 1, escribirá la letra "A"; si encuentra un 2, escribirá la letra "B" y finalmente, si encuentra el número 3, escribirá la letra "C"

De acuerdo con el enunciado del ejercicio en cada celda del rango A2:A6 se realizarán tres análisis: SI(A2=1 SI(A2=2 y SI(A2=3, respectivamente. Esto significa que hay que **anidar** tres funciones SI en la misma expresión. Afortunadamente Excel permite anidar hasta 7 funciones SI.

1. Inserten una nueva Hoja de Excel, la quinta, y transcriban los siguientes datos, tal y como se indica en la columnas A y B, respectivamente:

	A	B
1	Valores	Acciones
2	3	
3	2	
4	1	
5	3	
6	1	

2. Hagan clic en la celda B2 para activarla y escriban manualmente la siguiente función SI.

NOTA: Se deben escribir tantos paréntesis de cierre como funciones SI se aniden

=SI(A2=1;"A";SI(A2=2;"B";SI(A2=3;"C")))

3. Opriman la tecla **Enter**
4. Lleven el cursor nuevamente sobre la celda **B2**
5. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **B3:B6**. Actualicen el libro.

La función inspeccionó inteligentemente los valores almacenados en el rango A2:A6 y actuó tal y como se le había solicitado a la función SI que hiciera

	A	B
1	Valores	Acciones
2	3	C
3	2	B
4	1	A
5	3	C
6	1	A

Ejemplo F:

En el siguiente ejemplo se utilizará la función SI con el fin de calificar las notas obtenidas por los alumnos de la asignatura Computación, en el semestre B-2000. Para ello se usarán las siguientes Distinciones utilizados en el sistema educativo Venezolano: **Sobresaliente**, 20 puntos; **Distinguido**, 19; **Muy bueno**, 17-18; **Bueno**, 16; **Regular**, 11-15; **Aprobado**, 10 y **Reprobado**, 0 - 9, respectivamente.

Tal como puede observarse existen 7 categorías de calificaciones, por lo tanto se deben **anidar** siete funciones SI en la misma expresión. El anidamiento debe hacerse en el mismo orden en que aparecen las categorías, es decir desde Sobresaliente hasta Reprobado (o viceversa).

1. Inserten una nueva Hoja de Excel, la sexta y transcriban los siguientes datos, tal y como se indica en la columnas A, B, C, D, E y F, respectivamente:

	A	B	C	D	E	F
1	N. Promedio	Apellidos	Nombres	Escuela	Nota	Categoría
2						
3	13.38	Albornoz	Pedro	EC	13	
4	8.00	Araujo	Benito	EC	12	
5	14.50	Barrios	Isabel	EC	20	
6	15.79	Cabrera	Antonio	EA	18	
7	13.55	Carrillo	Luis	EC	11	
8	9.62	Chacón	Domingo	EC	14	
9	11.73	Contreras	José	EC	16	
10	15.33	Dávila	Ernesto	EA	17	
11	14.38	Durán	Petra	EA	13	
12	16.21	Espinoza	Luisa	EC	14	
13	14.00	Ferrer	Antulio	EC	15	
14	8.00	Lobo	Héctor	EC	10	
15	5.37	López	Ibrahín	EC	10	
16	16.86	Pérez	Josefa	EA	20	
17	8.71	Pérez	Rafael	EC	12	
18	13.73	Prieto	Juan	EC	19	
19	18.33	Robledo	Antonia	EC	12	
20	11.33	Toro	Rebeca	EC	15	
21	10.48	Toro	Alicia	EC	8	
22	14.20	Troconis	Elizabeth	EC	12	
23	8.41	Valero	Faustino	EC	11	
24	7.53	Zambrano	Yuly	EC	10	

2. Hagan clic en la celda F3 para activarla y escriban la siguiente función SI. **NOTA:** Escriban tantos paréntesis de cierre como funciones SI aniden:

```
=SI(E3=20;"Sobresaliente";SI(E3>=19;"Distinguido";  
SI(E3>=17;"Muy bueno"; SI(E3=16;"Bueno";SI(E3>=11;  
"Regular";SI(E3=10;"Aprobado; SI(E3<10;"Reprobado"  
))))))
```

3. Opriman la tecla **Enter**
4. Lleven el cursor nuevamente sobre la celda **F3**
5. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **F3:F24**. Actualicen el libro.

La función inspeccionó inteligentemente los valores almacenados en el rango E3:E24 y actuó tal y tal y como se especificó en la función SI

Resultados:

	A	B	C	D	E	F
	Av.	Apellidos	Nombres	Escuela	Nota	Mención
1						
2						
3	13.38	Albornoz	Pedro	EC	13	Regular
4	8.00	Araujo	Benito	EC	12	Regular
5	14.50	Barrios	Isabel	EC	20	Sobresaliente
6	15.79	Cabrera	Antonio	EA	18	Muy bueno
7	13.55	Carrillo	Luis	EC	11	Regular
8	9.62	Chacón	Domingo	EC	14	Regular
9	11.73	Contreras	José	EC	16	Bueno
10	15.33	Dávila	Ernesto	EA	17	Muy bueno
11	14.38	Durán	Petra	EA	13	Regular
12	16.21	Espinoza	Luisa	EC	14	Regular
13	14.00	Ferrer	Antulio	EC	15	Regular
14	8.00	Lobo	Héctor	EC	10	Aprobado
15	5.37	López	Ibrahin	EC	10	Aprobado
16	16.86	Pérez	Josefa	EA	20	Sobresaliente
17	8.71	Pérez	Rafaél	EC	12	Regular
18	13.73	Prieto	Juán	EC	19	Distinguido
19	18.33	Robledo	Antonia	EC	12	Regular
20	11.33	Toro	Rebeca	EC	15	Regular
21	10.48	Toro	Alcia	EC	8	Reprobado
22	14.20	Troconis	Elizabeth	EC	12	Regular
23	8.41	Valero	Faustino	EC	11	Regular
24	7.53	Zambrano	Yuly	EC	10	Aprobado

Ejemplo G:

En el siguiente ejemplo, si el valor en la celda A2 es 100, la prueba lógica será Verdadero y se calculará la suma total del rango A2:A7. De lo contrario, si la prueba_lógica es Falso, se devolverá una **cadena de texto vacía** ("") que borrará el contenido de la celda que contenga la función SI

Prueba_lógica	A2=100
Valor_si_verdadero	Sumatoria del rango
Valor_si_falso	Cadena de texto vacía ("")

1. Inserten una nueva Hoja de Excel, la séptima y transcriban los siguientes datos, tal y como se indica en la columnas A y B, respectivamente:

	A	B
1	Valores	Suma
2	100	
3	80	
4	100	
5	100	
6	88	
7	100	

2. Hagan clic en la celda B2 para activarla y escriban manualmente la siguiente función SI.

=SI(A2=100;SUMA(\$A\$2:\$A\$7);"")

3. Opriman la tecla **Enter**
4. Lleven el cursor nuevamente sobre la celda **B2**
5. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **B3:B7**. Actualicen el libro.

La función inspeccionó inteligentemente los valores almacenados en el rango A2:A7 y actuó tal y como se le había solicitado a la función SI

	A	B
1	Valores	Suma
2	100	568
3	80	
4	100	568
5	100	568
6	88	
7	100	568

Ejemplo H:

Suponga que una hoja de cálculo contiene las cifras de los gastos actuales y los pronosticados. Las celdas B4:B6 contienen los **Gastos actuales** para enero, febrero y marzo: 1500, 500 y 5000, respectivamente. Las celdas del rango C4:C6 contienen los **Gastos pronosticados** para los mismos meses: 900, 900 y 925.

Prueba_lógica	B4>C4
Valor_si_verdadero	Presupuesto excedido
Valor_si_falso	Aceptar

1. Inserten una nueva Hoja de Excel, la octava y transcriban los siguientes datos, tal y como se indica en la columnas A, B, C y D, respectivamente:

	A	B	C	D
1	Gastos		Gastos	
2	Meses	actuales	Pronosticados	
3				
4	Enero	1500	900	
5	Febrero	500	900	
6	Marzo	500	925	
7				

2. Hagan clic en la celda D4 para activarla y escriban manualmente la siguiente función SI.

=SI(B4>C4;"Presupuesto excedido";"Aceptar")

3. Opriman la tecla **Enter**
4. Lleven el cursor nuevamente sobre la celda **D4**
5. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **D5:D6**. Actualicen el libro.

La función inspeccionó inteligentemente los valores almacenados en el rango D4:D6 y actuó tal y como se había previsto en la función SI

	A	B	C	D
1	Gastos		Gastos	
2	Meses	actuales	Pronosticados	
3				Exceso
4	Enero	1500	900	Presupuesto excedido
5	Febrero	500	900	Aceptar
6	Marzo	500	925	Aceptar
7				

La Función Lógica Y()

De acuerdo con el nivel de ayuda de Microsoft Excel, la función **=Y()** devolverá la acción **Verdadero** si todos los argumentos son **Verdadero** o la acción **Falso** si uno o más argumentos son **Falsos**.

Sintaxis:

Y(valor_lógico1;valor_lógico2; ... ;Valor_lógico30)

Esta función puede manejar hasta 30 condiciones que pueden ser **Verdadero** o **Falso**.

Ejemplos:

Si se cumplen todas las condiciones el resultado es verdadero:

- Y(Verdadero; Verdadero), entonces es igual a Verdadero
- Y(2+2=4; 2+3=5), entonces es igual a Verdadero
- Si B4 contiene un número entre 1 y 100, entonces Y(1<B4; B4<100) es igual a Verdadero

Si no se cumple una condición el resultado es Falso:

- Si B1:B3 contiene los valores Verdadero, Falso y Verdadero, entonces: Y(B1:B3) es igual a Falso
- Y(Verdadero; Falso), entonces es igual a Falso

Los ejemplos planteados en las páginas anteriores suponen que se debe cumplir una sola condición. Pero, ¿qué pasaría si se tuvieran que cumplir más de una condición ?.

Supongan que la función **=SI** tenga que tomar en cuenta las **dos** condiciones.

- Que la función **=SI** hiciese algo sólo si se tuvieran que cumplir las **2 condiciones**
- Que la función **=SI** hiciese algo si se cumpliera **una de las dos condiciones**

Estas dos condiciones se pueden controlar mediante los operadores lógicos: **Y** y **O**, respectivamente:

La sintaxis de cada orden es la siguiente:

=SI(Y(Condición1:Condición2...)

La letra **Y** controla que se cumplan las dos condiciones

=SI(O(Condición1:Condición2..)

La letra **O** controla que se cumpla una de las dos condiciones

Ejemplo I:

Se utilizará la función SI con el fin de conocer cuantos estudiantes de la Escuela de Contaduría (**primera condición**); con promedio en la carrera de 14 puntos o más (**segunda condición**) que hayan aprobado Computación con mas de 14 puntos (**tercera condición**), pudieran obtener una nota definitiva equivalente al promedio de estas dos últimas notas, mientras el resto del curso obtendría nota definitiva igual al promedio de las notas de la asignatura Computación.

Prueba_lógica	Y(A3>=14;D3="EC";E3>=14)
Valor_si_verdadero	Promedio(A3;E3)
Valor_si_falso	Promedio(\$E\$3:\$E\$24)

Los datos para este problema son los mismos que aparecen en la página 7

1. Inserten una nueva Hoja de Excel, la novena y transcriban los siguientes datos, tal y como se indica en la columnas A, B, C, D, E y F, respectivamente.
2. Hagan clic en la celda F3 para activarla y escriban manualmente la siguiente función =SI:

=SI(Y(A3>=14;D3="EC";E3>=14);Promedio(A3;E3);Promedio(\$E\$3:\$E\$24))

Esta función pareciera decir:

Si se cumplen las 3 condiciones simultáneamente: (**A3**>=14;D3="EC";E3>=14), la función Si calculará el promedio de las notas almacenadas en las celdas A3:E3 y colocará el resultado en las celdas de aquellos estudiantes que cumplan dichas condiciones. Por el contrario, si no se cumple una de las tres condiciones, entonces la función **Verdadero** calculará el promedio del rango E3:E24 y se colocará el resultado en las restantes celdas del rango.

3. Opriman la tecla **Enter**
4. Lleven el cursor nuevamente sobre la celda **F3**
5. Coloquen el cursor sobre el **controlador de relleno** y hagan doble clic sobre él para copiar la fórmula en el resto del rango, es decir: **F3:F24**. Actualicen el libro.

A continuación se muestran las notas de los primeros 7 estudiantes de la lista

	A	B	C	D	E	F
	Av.	Apellidos	Nombres	Escuela	Nota	Mención
1						
2						
3	13.38	Albornoz	Pedro	EC	13	13.63636364
4	8.00	Araujo	Benito	EC	12	13.63636364
5	14.50	Barrios	Isabel	EC	20	17.25
6	15.79	Cabrera	Antonio	EA	18	13.63636364
7	13.55	Carrillo	Luis	EC	11	13.63636364
8	9.62	Chacón	Domingo	EC	14	13.63636364

Función =O()

De acuerdo con el nivel de ayuda de Microsoft Excel, la función **=O()** devolverá **Verdadero** si alguno de los argumentos es **Verdadero**; devolverá **Falso** si todos los argumentos son **Falso**.

Sintaxis

Y(valor_lógico1;valor_lógico2; ... ;Valor_lógico30)

Esta función puede manejar hasta 30 condiciones que pueden ser **Verdadero** o **Falso**.

Ejemplos:

Si se cumple uno de los argumentos, como mínimo, el resultado es verdadero:

- O(verdadero) es igual Verdadero
- Si el rango A1:A3 contiene los valores Verdadero, Falso y Verdadero, entonces O(A1:A3) es igual a Verdadero

Si no se cumplen las argumentos el resultado es Falso:

- Si O(1+1=1;2+2=5) es igual a Falso

En el cuadro No. 1 se presenta la relación de existencias de la empresa XYZ correspondiente al mes de julio del año en curso:

Cuadro 1. Relación de existencias y Stock Mínimo de la Empresa XYZ

Productos	Existencia al Inicio del mes	Cantidad Vendida	Existencias al Final de mes	Stock Mínimo	Acción
A123	2500	1350	1150	1500	
ABCD	3200	2500	700	500	
DC34	150	125	25	60	
NM90	225	100	125	150	
CODA	1450	1000	450	300	
CODB	647	600	47	100	
CODC	1300	500	800	1000	
CODD	560	100	460	300	

Se pide:

1. Transcriba el ejercicio a una hoja de cálculo de Excel
2. Analice, con la ayuda de la función **=SI**, la posibilidad de "Reponer el Inventario" o "No reponerlo", en un todo de acuerdo con el siguiente criterio:

Si las **Existencias al final del mes** son menores que el **Stock mínimo**, vaya a la columna **Acción** y escriba la frase "**Reponer el inventario**", de lo contrario escriba "**No reponer**".

3. Escriba en su hoja de examen la expresión **Si** utilizada con el fin de considerar válida la respuesta

